Plans tangents…un début

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en janvier 2006. L'objectif de cette feuille Maple est de présenter une approche intuitive donnant une équation d'un plan tangent à une surface donnée sur la base seule de la continuité des dérivées partielles. Rappelons que la continuité des dérivées partielles est une condition suffisante pour l'existence d'un plan tangent.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(plots,contourplot,display,spacecurve,setoptions,setoptions3d):
  with(plottools,point):
  macro(grispâle = COLOR(RGB, .9523, .9523, .9523)):
  setoptions3d(size=[300,300],axesfont=[TIMES,ROMAN,8],labelfont=
  [TIMES,ROMAN,8]);
  setoptions(size=[300,300],axesfont=[TIMES,ROMAN,8],labelfont=[TIMES,
  ROMAN,8]);
```

Plans tangents

Soit la fonction f définie par $f(x, y) = -2x^2 - y^2$.

```
> f:=(x,y)->-2*x^2-y^2;
```

$$f := (x, y) \mapsto -2 \cdot x^2 - y^2 \tag{2.1}$$

Obtenons le tracé de ce paraboloïde sur un pavé rectangulaire où $x \in [-1, 2]$ et $y \in [-1, 2]$ et illustrons sur la surface le point P(1,1,3).

```
> Surface:=plot3d([x,y,f(x,y)],x=-1..2,y=-1..2,
          grid=[25,25],style=patch,shading=zgrayscale,
          axes=framed,linestyle=0):
  P:=point([0,1,f(0,1)],color="Niagara 16", symbol=solidsphere):
> display(Surface,P,orientation=[55,75,0]);
```



Rappelons la forme scalaire de l'équation d'un plan passant par un point $P(x_0, y_0, z_0)$ et ayant un vecteur normal (A,B,C)

$$A(x-x_0) + B(y-y_0) + C(z-z_0) = 0$$

En divisant par C ($C \neq 0$) et avec un changement de notation, on a

$$z - z_0 = a(x-x_0) + b(y-y_0) \quad (1)$$

Si cette équation représente un plan tangent au point $P(x_0, y_0, z_0)$, l'intersection de ce plan avec le plan $y = y_0$ donne l'équation

$$z - z_0 = a(x-x_0) \quad \text{car} \quad y = y_0$$

On reconnaît cette équation comme l'équation d'une droite. Cette droite ne peut qu'être que la tangente dans ce plan sécant passant par le point P. Or, cette droite est de pente $a = f'_x(x_0, y_0)$.

De même, si on pose $x = x_0$ dans l'équation (1), on obtient $b = f'_y(x_0, y_0)$.

Donc, une équation du plan tangent au point $P(x_0, y_0, z_0)$ à une surface d'équation $z = f(x, y)$ est

$$z - z_0 = f'_x(x_0, y_0)(x-x_0) + f'_y(x_0, y_0)(y-y_0)$$

Rappelons que la continuité des dérivées partielles est une condition suffisante pour l'existence d'un plan tangent et non pas l'existence des dérivées partielles au point P.

```
> Éq_plan_tangent:=z-z0='D[1](f)'(x0,y0)*(x-x0)+'D[2](f)(x0,y0)'*(y-y0)
  ;
```

$$Éq\_plan\_tangent := z - z0 = \mathrm{D}_1(f)(x0, y0)(x - x0) + \mathrm{D}_2(f)(x0, y0)(y - y0) \quad \textbf{(2.2)}$$

Obtenons une équation du plan tangent à la surface d'équation $z = -2x^2 - y^2$ au point P(1,1,-3)

```
> x0:=0:
  y0:=1:
  z0:=f(x0,y0):
```

```
> Éq_plan_tangent;
```

$$z + 1 = -2y + 2 \quad \textbf{(2.3)}$$

Illustrons dans un même graphique la surface ainsi que le plan tangent.

```
> solve(Éq_plan_tangent,{z});
  assign(%);
```

$$\{z = -2y + 1\} \quad \textbf{(2.4)}$$

```
> Plan:=plot3d([x,y,z],x=-1..1,y=-0..2,style=patch,color="Niagara 1",
  transparency=0.5):
```

```
display(Surface,Plan,P,orientation=[55,75,0]);
#z:='z':
```



Zoomons vers l'avant sur le point P le graphique de la surface et du plan tangent.

```
> Surface:=plot3d([x,y,f(x,y)],x=-0.2..0.2,y=0.7..1.25,
          grid=[25,25],style=patch,color=grispâle,
          axes=framed,linestyle=0,glossiness=0):
  Plan:=plot3d([x,y,z],x=-0.2..0.2,y=0.8..1.2,style=patch,color=
  "Niagara 1",transparency=0.5):
  display(Surface,Plan,P,orientation=[55,75,0]);
```



Zoomons encore davantage.

```
> Surface:=plot3d([x,y,f(x,y)],x=-0.05..0.05,y=1.21..0.85,
          grid=[25,25],style=patch,color=grispâle,
          axes=framed,linestyle=0,glossiness=0):
  Plan:=plot3d([x,y,z],x=-0.05..0.045,y=1.2..0.85,style=patch,color=
  "Niagara 1",transparency=0.6):
  display(Surface,Plan,P,orientation=[55,75,0]);
```



Ce qu'on observe, c'est que plus on regarde de près la surface au point P, plus la surface du paraboloïde semble plate et ressemble à son plan tangent. On dit que la surface se "linéarise".

Une autre manière de constater ce fait est de voir comment se comporte les courbes de niveau avec ces différents zooms.

```
> contourplot(f(x,y),x=-1..2,y=-1..2,contours=12);
```



```
> contourplot(f(x,y),x=0.5..1.5,y=0.5..1.5,contours=12);
```




```
> contourplot(f(x,y),x=0.85..1.15,y=0.85..1.15,contours=12);
```



Les courbes de niveau présente de plus en plus une formation de droites parallèles équidistantes, ce qui est caractéristique des courbes de niveau d'un plan.